

# ZATITZAILE KOMUNETAKO HANDIENA ETA MULTIPLO KOMUNETAKO TXIKIENA

---

**D**uela bi ale, 31. alean hain zuzen, bide bati eman genion hasiera: Zenbakien FaktORIZAZIOARI eta zenbakia faktoretan edukita burutu daitezkeen eragiketak azaltzearen bideari. Ale honetan, Zatitzaile Komunetako Handiena eta Multiplo Komunetako Txikiena nola kalkulatzaren diren azaltzen da.

Kasu honetan erabiltzaileak emaitza zehatzak sartu behar ditu; Z.K.H. eta M.K.T.ren balioak alegia. Emaitza hauek kontrolatzea, besterik gabe, ez litzateke lan zaila izango ordenadorearentzat, baldin eta zenbakiak eta emaitzak aldezturik definituta egongo balira. Baina orain, gehienetan bezala, aleatoriotasuna sartu nahi izan dugu; erabiltzeko unean programa atseginagoa izan dadin. Beraz, programa berak kalkulatuko ditu Z.K.H. eta M.K.T.

GOGORATUZ!!!

## Zatitzaile Komunetako Handiena

Zenbaki arrunt batzuek aldi berean zatitzaileztaduten zenbakirik handiena.

## Multiplo Komunetako Txikiena

Zenbaki arrunt batzuk emanik, aldi berean zenbaki guztien multiploen arteko txikiena.

ADIBIDEA:

$$864 = 2^5 \cdot 3^3$$

$$909 = 3^2 \cdot 101$$

$$Z.K.H. = 3^2 = 9$$

$$M.K.T. = 2^5 \cdot 3^3 \cdot 101 = 87264$$

Hona hemen programa:

```
10 REM ZKH ETA MKT
20 DIM Z1(10,2),Z2(10,2),DES(10,2),ZKH(10,2),MKT(10,2)
30 SCREEN 2:SCREEN 0:KEY OFF
40 RANDOMIZE TIMER
50 GEH$="B"
60 WHILE GEH$="B" OR GEH$="b"
70 FOR I=1 TO 10:Z1(I,1)=0:Z1(I,2)=0:Z2(I,1)=0:Z2(I,2)=0:NEXT I
80 CLS:COLOR 29:LOCATE 12,32:PRINT "ITXARON ARREN!!!"
90 ALE1=INT(RND*999)+1:ZEN=ALE1:GOSUB 850
100 ZK1=J:FOR I=1 TO 10:Z1(I,1)=DES(I,1):Z1(I,2)=DES(I,2):NEXT I
110 ALE2=INT(RND*999)+1:ZEN=ALE2:GOSUB 850
120 ZK2=J:FOR I=1 TO 10:Z2(I,1)=DES(I,1):Z2(I,2)=DES(I,2):NEXT I
130 REM ZKHren kalkulua
140 FOR I=1 TO 10 :ZKH(I,1)=0:ZKH(I,2)=0:MKT(I,1)=0:MKT(I,2)=0:NEXT I
150 REM ZKHren kalkulua
```

```

160      I=0:J1=1:J2=1
170      WHILE J1<=ZK1
180          WHILE (Z1(J1,1)>Z2(J2,1)) AND Z2(J2,1)<>0:J2=J2+1:WEND
190          IF Z1(J1,1)<>Z2(J2,1) THEN 220
200              I=I+1:ZKH(I,1)=Z1(J1,1)
210              IF Z1(J1,2)<=Z2(J2,2) THEN ZKH(I,2)=Z1(J1,2)
                  ELSE ZKH(I,2)=Z2(J2,2)
220              J1=J1+1:J2=1
230          WEND
240          ZKH=1:ZK3=I
250          FOR JJ=1 TO I:ZKH=(ZKH(JJ,1)^ZKH(JJ,2))*ZKH:NEXT JJ
260      REM MKTren kalkulua
270      I=0:J1=1:J2=1
280      WHILE J1<=ZK1 AND J2<=ZK2
290          IF Z1(J1,1)>=Z2(J2,1) THEN 310
300          I=I+1:MKT(I,1)=Z1(J1,1):MKT(I,2)=Z1(J1,2):J1=J1+1:GOTO 360
310          IF Z1(J1,1)>Z2(J2,1) THEN 350
320          I=I+1:MKT(I,1)=Z1(J1,1)
330          IF Z1(J1,2)>Z2(J2,2) THEN MKT(I,2)=Z1(J1,2)
                  ELSE MKT(I,2)=Z2(J2,2)
340          J1=J1+1:J2=J2+1:GOTO 360
350          I=I+1:MKT(I,1)=Z2(J2,1):MKT(I,2)=Z2(J2,2):J2=J2+1
360      WEND
370      FOR J=J1 TO ZK1
380          I=I+1:MKT(I,1)=Z1(J,1):MKT(I,2)=Z1(J,2)
390      NEXT J
400      FOR J=J2 TO ZK2
410          I=I+1:MKT(I,1)=Z2(J,1):MKT(I,2)=Z2(J,2)
420      NEXT J
430      MKT=1:ZK4=I
440      FOR JJ=1 TO I:MKT=(MKT(JJ,1)^MKT(JJ,2))*MKT:NEXT JJ
450      REM AMAIA ZKH eta MKTren kalkulua
460      CLS:COLOR 10:LOCATE 1,31:PRINT "Z.K.H. ETA M.K.T.":COLOR 11:LOCATE 2,31:PRINT
STRING$(17,61)
470      LOCATE 19,5:COLOR 10:PRINT ALE1;"eta ";ALE2;"zenbakien Z.K.H.":COLOR 12:INPUT " ";EM1$
480      LOCATE 20,5:COLOR 10:PRINT ALE1;"eta ";ALE2;"zenbakien M.K.T.":COLOR 12:INPUT " ";EM2$
490      LOCATE 23,20:COLOR 3
500      IF ZKH=VAL(EM1$) AND MKT=VAL(EM2$) THEN
          PRINT "ONGI!!! Ongi burutu dituzu eragiketak";
          ELSE PRINT "GAIZKI!!! Gaizki burutu dituzu eragiketak";
510      FOR JJ=1 TO 3000:NEXT JJ
520      COLOR 13:LOCATE 6,9:PRINT ALE1;:LOCATE 6,14:PRINT "=";:COLOR 11:LOCATE 6,16
530      FOR II=1 TO ZK1
540          LOCATE CSRLIN,POS(0):PRINT Z1(II,1);
550          IF Z1(II,2)>1 THEN LOCATE CSRLIN-1,POS(0)-1:PRINT Z1(II,2);
560          IF Z1(II+1,1)>0 THEN LOCATE 6,POS(0):PRINT " ";
570      NEXT II
580      COLOR 13:LOCATE 9,9:PRINT ALE2;:LOCATE 9,14:PRINT "=";:COLOR 11:LOCATE 9,16
590      FOR II=1 TO ZK2
600          LOCATE CSRLIN,POS(0):PRINT Z2(II,1);
610          IF Z2(II,2)>1 THEN LOCATE CSRLIN-1,POS(0)-1:PRINT Z2(II,2);
620          IF Z2(II+1,1)>0 THEN LOCATE 9,POS(0):PRINT " ";
630      NEXT II
640      COLOR 13:LOCATE 12,10:PRINT "ZKH = ";:COLOR 11:LOCATE 12,16
650      FOR II=1 TO ZK3
660          LOCATE CSRLIN,POS(0):PRINT ZKH(II,1);
670          IF ZKH(II,2)>1 THEN LOCATE CSRLIN-1,POS(0)-1:PRINT ZKH(II,2);
680          IF ZKH(II+1,1)>0 THEN LOCATE 12,POS(0):PRINT " ";
690      NEXT II
700      LOCATE 12,POS(0):IF ZKH>1 THEN PRINT "=";
710      PRINT ZKH
720      COLOR 13:LOCATE 15,10:PRINT "MKT = ";:COLOR 11:LOCATE 15,16
730      FOR II=1 TO ZK4
740          LOCATE CSRLIN,POS(0):PRINT MKT(II,1);
750          IF MKT(II,2)>1 THEN LOCATE CSRLIN-1,POS(0)-1:PRINT MKT(II,2);

```

```

760          IF MKT(II+1,1)>0 THEN LOCATE 15,POS(0):PRINT ". ";
770      NEXT II
780      LOCATE 15,POS(0):PRINT "=";MKT
790      LOCATE 23,1:PRINT SPC(79)
800      COLOR 11:LOCATE 23,19:INPUT "Beste ariketarik egin nahi al duzu (B/E)?", GEH$
810  WEND
820  CLS
830  END
840
850  REM DESKONPOSATU
860      FOR I=1 TO 10:DES(I,1)=0:DES(I,2)=0:NEXT I
870      J=0:I=2
880      WHILE I<=INT(ZEN/2)
890          KONT=0
900          WHILE ZEN-INT(ZEN/I)*I=0
910              KONT=KONT+1:ZEN=INT(ZEN/I)
920          WEND
930          IF KONT>0 THEN J=J+1:DES(J,1)=I:DES(J,2)=KONT
940          I=I+1
950      WEND
960      IF J=0 THEN J=1:DES(J,1)=ZEN:DES(J,2)=1:GOTO 980
970      IF ZEN<>DES(J,1) AND ZEN>1 THEN J=J+1:DES(J,1)=ZEN:DES(J,2)=1
980      RETURN
990  REM AMAIA DESKONPOSATU

```

### Programaren azalpena

70-120: 1etik 999ra bitarteko bi zenbaki aleatorio sortu eta hauek faktoreetan deskonposatu.

130-250: Z.K.H.ren kalkulua. Bi zenbakiek dituzten faktore amankomunak hartzen dira berretzaile txikienarekin.

260-440: M.K.T.ren kalkulua. Bi zenbakietan agertzen diren faktore guztiak hartzen dira berretzaile handienarekin. Hots, errepikatzen direnak behin bakarrik.

460-510: Erabiltzaileak sartutako erantzunak jaso eta ONGI ala GAIZKI dauden egiaztatu eta berari adierazi.

520-780: Edozein kasutan, zenbakien faktORIZAZIOA eta emaitza zuzenak azaltzen dira pantailan.

850-990: Zenbaki bat faktoreetan deskonposatzeko prozedura. Ikus 31. aleko Etxeko Ordenadorea. ■

**EUSKAL  
ADITZ  
LAGUNTZAILEA**  
**(Softwarea)**

*Ordenadorea erabiliz,  
aditzarekin jolastuz, milaka  
ariketa egin ditzakezu*

**JADANIK  
KALEAN**